

5 平面曲線

第3回で関数のグラフのプロットについて学んだ。数式を手軽に視覚化できるのは Mathematica の強みの一つである。他にもプロットコマンドはあり、数式の形に応じて何を使えばよいかをまとめると次のようになる:


数式の形	例	コマンド
$f(x)$	$\sin x$	<code>Plot</code>
$f(x, y) = 0$	$x^2 + y^2 = 0$	<code>ContourPlot</code>
$f(x, y) > 0$	$x^4 + (x - 2y^2) > 0$	<code>RegionPlot</code>
$(x(t), y(t))$	$(t - \sin t, 1 - \cos t)$	<code>ParametricPlot</code>
$r = f(\theta)$	$r = \sin 4\theta$	<code>PolarPlot</code>

今回はこれらを使って平面曲線の表示法を学ぶ。早い話が数式を使ったお絵かきなので、あまり難しく考えずにコマンドを入力して画面に表示される曲線を眺めて楽しんでほしい。

5.1 陰関数表示された曲線


方程式 $x^2 + y^2 = 1$ を満たす点 (x, y) の集合が円周になることはよく知られている。このように x, y の関係式に曲線を定めることを陰関数表示いう。曲線が陰関数表示されている場合 y について解けば普通の関数になるはずだが、円周の場合 $y = \pm\sqrt{1-x^2}$ のように複数の関数に分かれてしまったり、そもそも方程式が明示的に解けない場合も多い。

陰関数表示された曲線は `ContourPlot` を用いて直接描くことができる。さっそく円周を描いてみよう。方程式 $f(x, y) = g(x, y)$ で定まる曲線を $a \leq x \leq b, c \leq y \leq d$ の範囲で描画するには `ContourPlot[f==g, {x, a, b}, {y, c, d}]` とすればよい (方程式なので等号 2 つ == になることに注意)。



複数の方程式を同時に表示したければリスト化すればよい (すなわちカンマで区切って並べて {} に入れる)。

```
ContourPlot[{x^2+y^2==1,(x-1)^2+y^2==1}, {x,-1,2}, {y,-1,1},
  AspectRatio -> Automatic]
```



ContourPlot も Plot と同様にオプション AspectRatio->Automatic をつけることができる. 指定しない場合は図の縦横比が 1 になる.


楕円の方程式は $\frac{x^2}{a^2} + \frac{y^2}{b^2} = 1$ であった (図省略).

```
ContourPlot[x^2/9 + y^2/4 == 1, {x, -3, 3}, {y,-2,2}, AspectRatio -> Automatic]
```


描画範囲は少し考えれば分かるし, 分からなくても数字を適当に変えていい具合になるまで試せばよい.

英単語の contour は “等高線” という意味であり, 地形図に見られるような等間隔にとった等高線を描くこともできる. 方程式の代わりに関数を書けばよい.

```
ContourPlot[x^2+y^2, {x,-1,1}, {y,-1,1}]
```




この図が描いているのは, 下の図のように 2 変数関数 $z = f(x, y)$ を曲面グラフで表したときに, z の値を等間隔にとってできる断面曲線の集合である :



5.2 不等式で表わされる領域

前節でみたように 2 変数の方程式は曲線を表すが、これを 2 変数の不等式にすると平面領域を表す。不等式で表わされる平面領域を表示するには `RegionPlot` を使えばよい。 $f(x, y) > 0$ で表わされる平面領域を $a \leq x \leq b$, $c \leq y \leq d$ の範囲で表示するには `RegionPlot[f>0, {x,a,b}, {y,c,d}]` とすればよい。

例えば $x^4 + (x - 2y^2) > 0$ で定まる平面領域は次のようになる:



図としては変化しないが、不等号として \leq , \geq を使いたければ `<=`, `>=` とすればよい。また境界を含まないことを強調したいのなら、たとえばオプション `BoundaryStyle->Dashed` などを付けてみてはいかがだろうか。

5.3 パラメータ表示された曲線

次のようにパラメータ (あるいは媒介変数) を用いて曲線を表すことができる:

$$c(t) = (f(t), g(t)), \quad a \leq t \leq b$$


これは, 座標平面にペンで曲線を描くところを想像したとき, 時刻 t におけるペン先の位置が $(f(t), g(t))$ にあると思えばよい.

これを Mathematica で描かせるには

```
ParametricPlot[{f(t),g(t)}, {t,a,b}]
```

とする. たとえばサイクロイド曲線 $c(t) = (t - \sin t, 1 - \cos t)$, $(0 \leq t \leq 2\pi)$ を描いてみよう.

```
ParametricPlot[{t-Sin[t],1-Cos[t]}, {t,0,2Pi}]
```




デフォルトで `AspectRatio->Automatic` となっている. また描画範囲を $p \leq x \leq q$, $r \leq y \leq s$ に限定したければオプション `PlotRange->{{p,q},{r,s}}` を加えればよい.

パラメータ表示された複数の曲線を同時に表示したい場合はやはりリスト化すればよい. つまり $\{f_1(t), g_1(t)\}$ と $\{f_2(t), g_2(t)\}$ を同じ t について同時に描くならば

```
ParametricPlot[{{f1(t),g1(t)}, {f2(t),g2(t)}}, {t,a,b}]
```

とすればよい. `{}` の対応に注意すること.

```
ParametricPlot[{{t-Sin[t],1-Cos[t]}, {t-Sin[t],1+Cos[t]}}, {t,0,2Pi}]
```




次の曲線はリサージュ図形とよばれるものである:

$$c(t) = (\cos(mt), \sin(nt + a)), \quad (0 \leq t \leq 2\pi)$$


ただし m, n は自然数, a は実数の定数とする. リサージュ (Lissajous) はフランス人の名前で, リサージュ図形と書くこともあるが綴りからするとリサージュが正しそうだ. 定数をたとえば $m = 3, n = 5, a = -\frac{3}{10}\pi$ とし曲線を描いてみよう:

ParametricPlot[{Cos[3t], Sin[5t-3Pi/10]}, {t, 0, 2Pi}]



オシロスコープという機械を知っているだろうか. 簡単にいうと2つの異なる正弦波 $f(t)$, $g(t)$ を x 軸, y 軸に入力してリサージュ図形を描く道具で, 図形を見て2つの波動の周波数(ここでの定数 m, n にあたるもの)の比を読み取ることができる. たとえば今表示した図を見てみると, 右端 ($x = 1$) のあたりに x 軸方向の山が3つ見られ, 上端 ($y = 1$) のあたりには y 軸方向の山が5つ見てとれる. このことから $m = 3$, $n = 5$ が測定される. a については $t = 0$ での座標 $(1, \sin a)$ から読み取れるはずで, この例の $-\frac{3}{10}\pi$ を肉眼で正確に測定するのは難しいが, よくみると小さい負の数であることぐらいは分かってもいい.

ここで問題. 次のリサージュ図形のグラフはどのように定数 m, n, a を定めればよいだろうか? 図形から値を推定し, Mathematica で描いて同じになるか試してみよう.




5.4 極座標表示された曲線

極座標における極方程式 $r = f(\theta)$, $a \leq \theta \leq b$ もまた曲線を表している. 原点から出て x 軸と θ の角をなす半直線を引いて, その上に原点からの距離が $f(\theta)$ になる点をプロットし, この作業を θ を変化させて曲線に相当する. (座標変換 $x = r \cos \theta$, $y = r \sin \theta$ をしてしまえば θ をパラメータとする曲線 $c(\theta) = (f(\theta) \cos \theta, f(\theta) \sin \theta)$ としても解釈できる.)

これも Mathematica で描いてみよう. 極座標表示された曲線 $r = f(t)$ を $a \leq t \leq b$ の範囲で描くには `PolarPlot[f[t], {t,a,b}]` を使う. (ただし Mathematica で入力しやすいよう変数 θ を t に変更した.)

螺旋(らせん) $r = \theta$, $(0 \leq \theta \leq 4\pi)$ を描くには次のようにする:


```
PolarPlot[t,{t,0,4Pi}]
```



`PolarPlot` においては自動的に `AspectRatio->Automatic` に設定される.

`PolarPlot` もリスト化により複数の曲線を同時に描くことができる.


```
PolarPlot[{t, -t}, {t, 0, 4 Pi}]
```



なぜこんな絵になるか式から理解できるだろうか. 極座標 (r, θ) の r は原点からの距離なので 0 以上の値のみを考えるのが自然だが, 負の数に対しては原点を通り超えて反対側の点を考えていることになっているのである.

最後に $r = \sin(n\theta)$, ($0 \leq \theta \leq 2\pi$) で表わされるバラ曲線 (**Rose**) で遊んでみよう. n にいろいろな数字をあてはめてグラフを描いてみよ. たとえば $n = 4$ のときは次のようになる.

```
PolarPlot[Sin[4t],{t,0,2Pi}]
```



ここで n に整数以外の値を代入したヒネクレ者がいたとしたらなかなかセンスのある人だと思う. ただし t の範囲には注意が必要で, たとえば $n = \frac{p}{q}$ ならば $0 \leq t \leq 2q\pi$ にするとよい ($\sin(\frac{p}{q}t)$ の周期が $2q\pi$, 極座標 (r, θ) の角度 θ の周期が 2π なのでその最小公倍数をとる). また n に無理数をいれると周期性がなくなるので, 十分大きい値にしておくといよい (答えがあるわけではないので, いろいろ試していい具合になればそれが正解).

```
PolarPlot[Sin[2/5t],{t,0,10Pi}]
```

```
PolarPlot[Sin[Sqrt[2]t],{t,0,30Pi}]
```

5.5 練習問題

問題 1~8 は数式を Mathematica によって図示せよ.

問題 1 $(x^2 + y^2)^2 = x^2 - y^2$ ($-1 \leq x \leq 1, -1 \leq y \leq 1$) [レムニスケートあるいは連珠形とよばれる.]

問題 2 $x^3 + y^3 = 6xy$ ($-4 \leq x \leq 4, -4 \leq y \leq 4$) [デカルトの正葉形 (folium)]

問題 3 $|x^2 + y| \leq |y^2 + x|$ ($-2 \leq x \leq 2, -2 \leq y \leq 2$)

問題 4 $c(t) = (\cos^3 t, \sin^3 t)$ ($0 \leq t \leq 2\pi$) [アステロイド]

問題 5 $c(t) = (\sin t, \cos t + \log(\tan \frac{t}{2}))$ ($0 < t < \pi$) [トラクトリクス (tractrix) あるいは曳航線とよばれる. 原点から y 軸に沿って上に進む船が, $(1, 0)$ の位置にある漂流物をひもで引っ張るときの漂流物の軌跡が描く曲線. ドイツ語で Hundkurve(犬の曲線) というそうだが, 犬のしつけがうまいドイツ人ならではのネーミング.]

問題 6 $r = 1 + \cos \theta$ ($0 \leq \theta \leq 2\pi$) [カーディオイド (Cardioid) とよばれ, 心臓形という意味.]


問題 7 (butterfly と fancy butterfly)

- $r = e^{\sin \theta} - 2 \cos(4\theta) \quad (0 \leq \theta \leq 4\pi)$
- $r = e^{\sin \theta} - 2 \cos(4\theta) + \sin^5 \frac{\theta}{12} \quad (0 \leq \theta \leq 20\pi)$

問題 8 (chrysanthemum curve すなわち菊曲線)

$$r = 5 \left(1 + \sin \left(\frac{11\theta}{5} \right) \right) - 4 \sin^4 \left(\frac{17\theta}{3} \right) \sin^8 (2 \cos(3\theta) - 28\theta) \quad (0 \leq \theta \leq 21\pi)$$

問題 9 次のリサージュ図形を描け. ただし $a = \frac{\pi}{3}$ である.



問題 10 Mathematica を使って面白い平面曲線を描け (面白い空間曲線や空間曲面でもよい).

※ 付録の後にも問題がある. 時間が余ったらやってみるとよい.

A 付録

Mathematica はほかにもたくさんのグラフィック機能を持っているので、ヘルプを参考に面白いコマンドを自ら発見してほしい。ここでは授業で紹介しきれなかった機能で重要と思われるものをいくつか説明する。

A.1 インタラクティブな操作

数式のプロットでしばらく遊んでみると、定数の値を少しずつ書き換えて図の変化を確かめる、といった作業を自然にするようになるのではないだろうか。そんな操作をツマミ一つで実現する夢のコマンドが `Manipulate` だ。例えば楕円 $\frac{x^2}{a^2} + \frac{y^2}{b^2} = 1$ の定数 a, b を 1 から 3 の範囲で動かしてみたいなら、次のようにすればよい:

```
Manipulate[ContourPlot[x^2/a^2+y^2/b^2==1, {x,-3,3}, {y,-3,3}, AspectRatio->Automatic],  
  {a, 1, 3},{b,1,3}]
```

画面中のレバーをマウスでドラッグして動かすと a の値が変化してそれに対応する楕円を描画する。

他のプロットコマンドに対しても、変化させたい定数を含む形でプロットコマンドを書き、それを `Manipulate` の中に入れればよい。

リサージュ図形の m, n, a を動かしたければ次のようにすればよい:

```
Manipulate[ParametricPlot[{Cos[m t], Sin[n t+a]}, {t,0,2Pi}, PlotRange->{{-1,1},{-1,1}},  
  {a, -Pi, Pi}, {n, 0, 10, 1}, {n, 0, 10, 1}]
```


ここで $\{n, 0, 10, 1\}$ は n を 0 から 10 まで 1 刻みで、すなわち 0 から 10 までの整数で動かすことを意味している。また、 a, m, n の値を変化させた図形を同じ状態で比較したいのなら、オプション `PlotRange` で表示範囲を固定したほうがよいだろう。

A.2 いたるところ微分不可能な連続関数: 高木関数

数学の授業で、「微分可能ならば連続だが、連続だからといって微分可能とは限らない」という事実を習う。授業では通常、 $f(x) = |x|$ のような“角のある”関数を例にあげるので、連続関数の微分不可能点は“不幸な事故”であるかのような印象を受けるかもしれない。ところが、連続でありながら、どこもかしこも微分不可能な関数が存在するといったら信じられるだろうか。また、その関数のグラフはどんな風に見えるだろうか。


その関数の構成法のヒントは、第 3 回の演習問題のギザギザグラフにある。少し形が違うが、次の関数 $f(x)$ を定義する。

```
f[x_] := Abs[x - Floor[x + 1/2]]  
Plot[f[x], {x, 0, 1}, AspectRatio -> Automatic]
```



これを第 0 段階とし、ギザギザを倍にして高さを半分にする、という操作を繰り返した第 k 段階のノコギリ型グラフを持つ関数は $\frac{f(2^k x)}{2^k}$ で得られる。これを `saw[x, k]` とおこう。

```
saw[x_, k_] := f[2^k x]/2^k
Manipulate[Plot[saw[x,k], {x,0,1}, PlotRange->{0,1}, AspectRatio->Automatic], {k,1,10,1}]
```




ここで、 k は整数値であることから、`Manipulate[...,{k,1,10,1}]` (k は 1 から 10 まで刻み幅 1 で) とした。こうすれば、微分不可能点が倍々が増えていくのだが、 $k \rightarrow \infty$ の極限をとると定数関数 0 に収束してしまう。そこでもう一工夫して、ノコギリ型関数を第 1 から第 k 段階まで足した関数

```
takagi[x_, k_] := Sum[saw[x, i], {i, 1, k}]
```

を定義する。ここで、`Sum[...,{i, 1, k}]` は $\sum_{i=1}^k$ を意味する。

```
Manipulate[Plot[takagi[x,k],{x,0,1},PlotRange->{0,1},AspectRatio->Automatic],{k,1,10,1}]
```



この $k \rightarrow \infty$ での極限、 $T(x) = \sum_{i=1}^{\infty} \frac{f(2^i x)}{2^i}$ は連続だがいたるところ微分不可能な関数になる。この関数は『解析概論』の著者である高木貞治によって 1903 年に考案され、高木関数とよばれている。もちろん Mathematica では有限の k を近似的に表示することしかできないが、おおよその雰囲気はつかめるであろう。


A.3 空間図形のプロット

Mathematica には空間図形のプロットコマンドもたくさんある。以下に挙げるものは平面のプロットから容易に類推できるであろう：

数式の形	例	コマンド
$f(x, y)$	$\sin(x^2 + y^2)e^{-x^2}$	Plot3D
$f(x, y, z) = 0$	$6x^2 - 2x^4 - y^2z^2 = 0$	ContourPlot3D
$f(x, y, z) > 0$	$x^4 + (x - 2y^2) - z > 0$	RegionPlot3D
$(x(t), y(t), z(t))$	$(\sin 3t, \cos 4t, \sin 5t)$	ParametricPlot3D


以下, 例を挙げることで説明としたい. まず Plot3D は 2 変数関数の曲面グラフのプロットを行う.

```
Plot3D[Sin[x^2 + y^2] Exp[-x^2], {x, -2, 2}, {y, -2, 2}]
```




3 変数関数の等位面 $f(x, y, z) = c$ は ContourPlot3D によって表わすことができる. ここで z 座標の範囲も指定していることに注意しよう.

```
ContourPlot3D[6 x^2 - 2 x^4 - y^2 z^2 == 0, {x, -5, 5}, {y, -5, 5}, {z, -5, 5}]
```




平面の場合と同様, 等式を不等式にするなら RegionPlot3D を使えばよい.

```
RegionPlot3D[x^4 + (x - 2 y^2) - z > 0, {x, -2, 2}, {y, -2, 2}, {z, -2, 2}]
```




パラメータ表示された曲線は `ParametricPlot3D` を使う.

```
ParametricPlot3D[{Sin[3 t],Cos[4 t],Sin[5 t]}, {t,0,2Pi}]
```



ただし、このままでは曲線が線で表わされているので空間的な配置が把握しづらい。そんな時はオプション `PlotStyle` を指定するとよいだろう。

```
ParametricPlot3D[{Sin[3 t],Cos[4 t],Sin[5 t]}, {t,0,2Pi}, PlotStyle->Tube[0.05]]
```




さらに Manipulate と組み合わせることもできる.

```
Manipulate[ ParametricPlot3D[{Sin[n t], Cos[4 m t], Sin[5 t]}, {t, 0, 2 Pi},
  PlotStyle -> Tube[0.05]], {n, 1, 10, 1}, {m, 1, 10, 1}]
```


A.4 作品集

次に挙げるのは練習問題 10 に対して、過去の受講生から寄せられた解答例である:


```
ParametricPlot[{Cos[3 t] + 1/10 Cos[30 t], Sin[5 t - 3 Pi/10] + 1/10 Sin[50 t]},
  {t, 0, 2 Pi}, AspectRatio -> Automatic]
```




```
PolarPlot[E^Cos[t] - Cos[6 t], {t, 0, 2 Pi}]
```




```
PolarPlot[{Sin[Sqrt[E] t], Cos[Sqrt[E] t], Tan[Sqrt[E] t]}, {t, 0, 15 Pi}]
```




```
ContourPlot[y==Sum[(1/2)^n Cos[12900000001^n Pi x], {n, 0, 23}], {x, -1, 1}, {y, -2, 2}]
```



```
ParametricPlot[{Cos[5 t] + Sin[3 t], Sin[3 t] + Cos[5 Pi]}, {t, 0, 2 Pi}]
```



```
PolarPlot[Exp[Sin[3 t]] - Abs[2 Cos[7 t]], {t, 0, 4 Pi}]
```



A.5 追加問題

問題 11 次の方程式で表わされる曲面を描け. 次頁の図を参考にして, 描画範囲は適当に指定すること.

1. $x^6 + y^6 + z^6 = 1$

2. $x^2 + y^2 + z^2 + 1000(x^2 + y^2)(x^2 + z^2)(y^2 + z^2) = 1$

3. $\left(x^2 + \frac{9}{4}y^2 + z^2 - 1\right)^3 - x^2z^3 - \frac{9}{80}y^2z^3 = 0$

4. $x^2 + y^2 + z^3 = z^2$

